
Abstract. Problems from the area of graph theory always served as
fruitful benchmarks in order to explore the performance of Answer Set
Programming (ASP) systems. A relatively new branch in graph theory is
concerned with so-called secure sets. It is known that verifying whether
a set is secure in a graph is already co-NP-hard. The problem of enu-
merating all secure sets thus is challenging for ASP and its systems. In
particular, encodings for this problem seem to require disjunction and
also recursive aggregates. In this paper, we provide such encodings and
analyze their performance using the Clingo system.

1 Introduction

The paradigm of Answer Set Programming (ASP) [5] has been shown to be very
effective for tackling computationally complex problems [1, 21, 22]. Its Guess &
Check approach allows for easy specification and often very efficient solving of
NP-complete problems in practice. By allowing disjunctive logic programs, ASP
can even capture problems up to the second level of the polynomial hierarchy
[9]. However, compared to the vast amount of benchmarks for problems in NP,
problems at the second-level of the polynomial hierarchy have not been inves-
tigated at a large scale (with some exceptions like [1], [4] or [17]). To tune the
performance of ASP solvers for these problems, the community requires further
benchmarks from this category.

In this paper, we consider secure sets, a relatively new concept from graph
theory that was introduced by Brigham et al. extending the weaker notion of de-
fensive alliances [6]. The question of alliances has several practical applications.
For instance, it arises naturally when asking for common relations between peo-
ple, possible coalitions of political parties, or companies sharing the same eco-
nomic interests. It is, however, not limited to these examples but can generally be
applied when a classification of objects with respect to some common property
is necessary [23].



The secure set problem asks for a set of vertices S in a graph G such that
for each subset X ⊆ S, |N [X]∩S| ≥ |N [X] \S| holds, where N [X] is the closed
neighborhood of X in G, i.e., the set X together with all vertices adjacent to
some vertex in X. It is known that checking if a given set S is a secure set of
G is co-NP-complete [20]. This leads to the following observations regarding an
encoding for enumerating all secure sets:

– Since the verification problem is already co-NP-hard, suitable encodings re-
quire the full power of ASP, in particular disjunction (unless NP = co-NP).

– Such encodings typically rely on a recursive schema (the so-called saturation
technique) for the verification part.

– From the definition of secure sets we can conclude that aggregates have to
be employed also in the recursive part.

Aggregates in ASP have been thoroughly investigated [12, 13, 19, 24]. Due to
various underlying semantics, the implementation of aggregates differs between
individual ASP systems. Moreover, for the sake of simplicity, some systems im-
pose restrictions on aggregates such that not everything that is syntactically
and semantically expressible can also be used in practice. Therefore, we give a
detailed account of how our desired encodings can be realized using Clingo 4,
one of the prominent ASP systems.

To summarize, our main contributions are as follows:

– We provide encodings for finding secure sets in graphs, a problem which to
the best of our knowledge has not been tackled with ASP yet. Our encodings
require the “full power” of ASP, i.e., disjunction (in combination with the
saturation technique) plus aggregates.

– We present several optimizations for our encodings and provide an experi-
mental analysis, thereby comparing the run-time performance of the encoded
variants.

– We additionally state theoretical observations in form of new characteriza-
tions for secure sets.

This paper is structured as follows. Section 2 covers the required background
on secure sets and ASP. In Section 3 we present our ASP encodings for the prob-
lems under consideration. Experimental results for these encodings are reported
in Section 4. Section 5 concludes the paper with a discussion of our results.

2 Preliminaries

In this section, we define the notions of secure sets and then give a brief intro-
duction to Answer Set Programming (ASP). Beside the basic ASP syntax and
semantics we further describe the concept of the saturation technique and the
commonly used extension of the language for aggregates.



2.1 Secure Sets

Let G = (V,E) be a simple graph with vertex set V and edge set E. The set of
vertices adjacent to a vertex v ∈ V , the open neighborhood of v, is denoted by
N(v). The closed neighborhood N [v] of a vertex v ∈ V is the open neighborhood
of v together with the vertex v itself, formally N [v] = N(v) ∪ {v}. For a set
S ⊆ V , N(S) =

⋃
v∈S N(v) and N [S] =

⋃
v∈S N [v] define the open and the

closed neighborhood of S.
A defensive alliance of G = (V,E) is a subset S ⊆ V such that for any x ∈ S

the inequality |N [x]∩S| ≥ |N [x]\S| is satisfied. For a better understanding, one
can think of the vertices of N [x]∩ S as the defenders of x and those of N [x] \ S
as the attackers of x. This means that for any vertex contained in a defensive
alliance there are at least as many defenders as there are attackers and so any
attack on a single vertex can be repelled.

A more restrictive concept than defensive alliances are secure sets [6] where
simultaneous attacks on more than a single vertex have to be defended. A non-
empty set S ⊆ V is said to be secure in G = (V,E) if and only if for each subset
X of S the inequality |N [X] ∩ S| ≥ |N [X] \ S| holds. In this paper we consider
the following enumeration problems:

Problem 1. Secure Set: Given a graph G = (V,E), what are the secure sets
S ⊆ V , where a set S is secure if for all X ⊆ S the inequality |N [X] ∩ S| ≥
|N [X] \ S| holds?

Problem 2. Minimum Secure Set: Given a graph G = (V,E), what are the
cardinality-minimal secure sets S ⊆ V ?

2.2 Answer Set Programming

ASP [5] is a declarative language where a program Π is a finite set of rules

a1| . . . |ak ← b1, . . . , bm,not bm+1, . . . ,not bn.

where a1, . . . , ak, b1, . . . , bn are atoms. The constituents of a rule r ∈ Π are its
head h(r) = {a1, . . . , ak}, and its body consisting of b+(r) = {b1, . . . , bm} and
b−(r) = {bm+1, . . . , bn}. Intuitively, r states that if an answer set contains all of
b+(r) and none of b−(r), then it contains some element of h(r). An interpretation
I is a subset of atoms over the domain. I satisfies a rule r iff I ∩ h(r) 6= ∅ or
b−(r)∩I 6= ∅ or b+(r)\I 6= ∅. I is a model of a set of rules iff it satisfies each rule.
I is an answer set of a program Π iff it is a subset-minimal model of the program
ΠI = {h(r)← b+(r) | r ∈ Π, b−(r) ∩ I = ∅}, called the Gelfond-Lifschitz reduct
of Π with respect to I [18].

ASP systems allow first-order atoms of the form p(t1, . . . , tl) where p is a
predicate of arity l ≥ 0 and each ti is either a variable or a constant. With
p(t1; . . . ; tj) we denote the sequence of unary atoms p(t1), . . . , p(tj). An atom
is ground if it is free of variables. For any program Π, let UΠ be the set of all
constants appearing in Π. For non-ground programs, the above semantics can
be utilized by first applying, to each rule r ∈ Π, all possible substitutions from
the variables in r to elements of UΠ .



The Saturation Technique. The saturation technique allows us to represent
problems on the second level of the polynomial hierarchy by encoding the co-
NP-check in ASP [10]. This technique relies on the fact that rule heads may
contain disjunctions. This way, all atoms that are subject to a guess can be
jointly contained in an answer set.

The idea how this step can be employed to compute solutions for problems
on the second level of the polynomial hierarchy is the following. In the first place,
we guess a solution candidate for which we want to know if all possibilities of
this being in fact no valid solution fail. Therefore, at the same time we also guess
(using disjunction) a potential witness of the solution candidate being invalid.
If this second guess indeed does not yield such a witness, we derive a designated
atom that causes all atoms in the disjunction for guessing witnesses to be set to
true. By doing so, all models not amounting to a witness are “saturated” with all
the atoms in this disjunction. So, if a valid solution has been guessed, all guesses
of potential witnesses thus collapse to a unique maximal answer set. On the other
hand, if we have managed to guess a witness, we kill the solution candidate by
means of a constraint. Invalid solution candidates are then discarded by the
minimal model semantics because each model of the program that does not
encode a witness is saturated and is thus not a minimal model of the reduct.
For more details, we refer the reader to [10].

A concept that is often applied in combination with saturation is that of
a “loop” (see, e.g., [11]). A loop allows to avoid (unstratified) default negation
on the saturated parts of the program by “iterating” over the witnesses to be
checked.

Aggregates. Beside default negation, also recursive (or saturation-dependent)
aggregates may impose non-obvious results on the ASP program. In particular,
different ASP solvers handle aggregates differently. In the following we give an
overview of the semantics of aggregates. In order to cope with simple arithmetic
operations, the basic ASP language is commonly extended by aggregate functions
(cf. the ASP-Core-2 input language [7] and, e.g., [12]).

Syntax. An aggregate element e is of the form

t1, . . . , tm : l1, . . . , ln

where t1, . . . , tm are terms (denoted by Term(e)) and l1, . . . , ln are atoms, (de-
noted by Conj(e)). An aggregate atom has the form

#aggr{e1; . . . ; en} ≺ t

where #aggr, called an aggregate function, is one of {#count, #sum, #max, #min},
e1, . . . , en are aggregate elements, ≺ ∈ {<,≤, >,≥,=, 6=} is an aggregate relation
and t is a term. ASP programs under this extension allow aggregate atoms in
rule bodies.

In a given program Π an aggregate atom #aggr{e1; . . . ; en} ≺ t is called
recursive if it involves a cyclic dependency, i.e., if it contains an aggregate element



e with an atom a ∈ Conj(e) and there is a rule r with an atom b ∈ h(r) such
that a and b cannot be stratified. Otherwise it is called non-recursive.

Example 1. The program

Π : {p(1)← #sum{X : p(X)} > 0}.

contains a recursive aggregate since the atom p(1) in the head of the rule depends
on an aggregate atom involving p(1). 4

Semantics. Aggregate functions are evaluated with respect to an interpretation
I. A straightforward semantics for aggregates was proposed by Dell’Armi et
al. [8] and Faber et al. [12]. In the following we describe the semantics for the
ground case. The valuation of an aggregate element e = 〈t1, . . . , tm : l1, . . . , ln〉
is defined by

I(e) =

{
t1 if I |= Conj(e)

0 otherwise,

i.e., the projection of Term(e) on the first term, if Conj(e) evaluates to true
under I, and 0 otherwise. An aggregate function #aggr{e1; . . . ; en} is now simply
evaluated as the application of the function denoted by #aggr on

⋃n
i=1 I(ei). If⋃n

i=1 I(ei) is not in the domain of #aggr, then I(#aggr{e1; . . . ; en}) = ⊥. Based
on these definitions, #aggr{e1; . . . ; en} ≺ t ∈ I holds iff

(i) I(#aggr{e1; . . . ; en}) 6= ⊥ and
(ii) I(#aggr{e1; . . . ; en}) ≺ t holds.

Note that when aggregates are involved, the reduct of a program Π under an
interpretation I is obtained by deleting the rules in which a body literal is false.
For programs restricted to the basic ASP language, the semantics when using
this notion of a reduct is equivalent to the semantics using the Gelfond-Lifschitz
reduct [12].

Example 2. Consider the program

Π = {q ← #sum{1 : q} = 1.},

and interpretations {}, {q}, for which we obtain:

Π{} = ∅, Π{q} = Π

Under the subset-minimal model semantics {} is the only answer set of Π, since
{q} is not minimal and thus not a valid answer set. 4

In contrast, a different way of defining the semantics of aggregates, which is
implemented in, e.g., Clingo, is by translating them to another formula which is
evaluated instead. Ferraris and Lifschitz proposed such a translation to evaluate
so-called weight constraints [15], which can also be used to define the semantics
of aggregates [13, 19]. In a nutshell, the aggregate is translated to a conjunction
of implications reflecting the intuitive meaning of the aggregate.



It has been shown that both semantics are equivalent, at least when con-
sidering positive programs [14]. This result suggests that we can provide an
encoding that can equivalently be run with DLV and Clingo. However, nega-
tive numbers within aggregates are neither allowed in the current version of
DLV (Dec 16, 2012) [3], nor in the WASP system (Jun 12, 2013) [2], which uti-
lizes a modified version of DLV. We therefore restrict ourselves to a realization
using Clingo (4.3.0) [16].

3 Encodings

As the definition of secure sets suggests, recursive aggregates are required to cope
with the co-NP-hard task of verifying whether a guessed set of vertices is secure.
In what follows, we give two proposals for such encodings. The main obstacle
here is to get the recursive aggregates to work. In Clingo, it is required that the
value that the outcome of the aggregate is compared to is fixed, and moreover
that negative parts in aggregates have to stem from “outside” the saturation, i.e.,
they are not allowed to be part of the recursion. Note that excluding (default-
negated) negative parts from the saturation is required in general.

3.1 Loop Encoding

In order to fulfill the requirements above, we give a slightly different characteri-
zation of secure sets:

Lemma 1. Let G = (V,E) be a graph. A set S ⊆ V is secure in G iff for all
X ⊆ S

|N [X] ∩ S|+ |(N [S] \N [X]) \ S| − |N [S] \ S| ≥ 0.

Proof. Recall that S is secure iff for all X ⊆ S

|N [X] ∩ S| ≥ |N [X] \ S|.

We now add |(N [S] \N [X]) \ S| to both sides. Note that J = (N [S] \N [X]) \ S
and J ′ = (N [X]\S) are obviously disjoint and that J ∪J ′ = N [S]\S. It follows
that |J |+ |J ′| = |N [S] \ S| and thus we obtain

|N [X] ∩ S|+ |(N [S] \N [X]) \ S| ≥ |N [S] \ S|

yielding the desired result. ut

Note that this characterization has a subtraction involved (|N [S] \ S|) but
this number is independent from X, the set which has to be dealt with in the
saturation part of the encoding.

We now take this result further by defining secure sets based on their border :

Definition 1. Let G = (V,E) and S ⊆ V . The border b(S) of S is b(S) = {x |
{x, y} ∈ E, x ∈ S, y /∈ S}, i.e., the set of all vertices in S which are adjacent to
at least one vertex y /∈ S.



Lemma 2. Let G = (V,E) be a graph. A set S ⊆ V with border b(S) is secure
in G iff for all X ⊆ b(S)

|N [X] ∩ S|+ |(N [S] \N [X]) \ S| − |N [S] \ S| ≥ 0.

Proof. Given a secure set S, since b(S) ⊆ S, the “only if” direction follows
directly from Lemma 1. As for the other direction: Since the only negative part
of the inequality (|N [S] \ S|) is independent of all x ∈ S\b(S), checking the
inequality for all X ⊆ b(S) suffices to verify that S is secure. ut

In the following we present the ASP encoding for enumerating all secure
sets of a given graph using the concept of loops. The input graph G = (V,E)
is specified by means of the predicates vertex/1 and edge/2. We assume that
for every input fact edge(x,y) also edge(y,x) is given. Furthermore, let the
predicates inf/1 (infimum), succ/2 (successor), and sup/1 (supremum) specify
the lexicographical ordering over all vertices. Note that this ordering can be
obtained in polynomial time by a positive logic program.

In Encoding 1, a (non-empty) set S, denoted by inS/1, is guessed1 (rules
1–3). Predicate attackSet/1 represents the vertices in N [S]\S (rule 4), and
border/1 the set b(S) (rule 5). Following Lemma 2, it suffices to guess all sets
X ⊆ b(S), denoted by inX/1. Vertices in S\X are given by outX/1 (rules 6–7).
Predicate defendSet/1 specifies the set N [X] ∩ S (rules 8–9).

Rules 10–17 define the set (N [S]\N [X])\S, that is, the set of inactive at-
tackers. Intuitively, an inactive attacker is an attacker (i.e., a vertex in N [S]\S)
that is not adjacent to any vertex in X. In order to obtain the set of inactive
attackers (without using default negation on saturation-dependent atoms), we
have to loop. To be more precise, to determine for a vertex u ∈ V whether
u ∈ (N [S] \N [X]) holds, we “loop” over all vertices v ∈ V and check if one of
the following conditions holds: (a) v ∈ S\X (rules 10,13); (b) v ∈ V \S (rules
11,14); or (c) v ∈ X, {u, v} /∈ E (rules 12,15). Finally, the predicate inactAtt/1

denotes the set of inactive attackers (rule 17).
The set X ⊆ S is defended if X satisfies the inequality given in Lemma 2

(rule 18). In case defended/0 is obtained, we saturate by setting all vertices in S
to both inX/1 and outX/1 (rules 19–20); On the other hand, if X is not defended
against the attackers, the answer set is removed (rule 21). In that case, due to the
minimal model semantics, no answer set containing S is returned. Specifically,
in case any X ⊆ S cannot be defended, S is not secure and is therefore not
returned as a solution.

Example 3. Figure 1 shows an example graph Gex with currently selected sets
S = {a, b, c, f} and X = {c}. The figure illustrates the corresponding defend
set N [X] ∩ S = {b, c}, attack set N [S]\S = {d, e, g, h} and the set of inactive
attackers (N [S]\N [X])\S = {e, g}. Since |{b, c}|+|{e, g}|−|{d, e, g, h}| = 2+2−
4 = 0 ≥ 0 holds, X is defended. For S to be secure, all X ⊆ b(S), i.e., X ⊆ {c, f}
1 We remark that this guess can alternatively be specified by using disjunction or a

so-called choice rule (see, e.g., [7]), which we omit here for legibility reasons.



have to be defended. For X = {c, f}, we have N [X] ∩ S = {b, c, f}, N [S]\S =
{d, e, g, h} and (N [S]\N [X])\S = {}. Here, |{b, c, f}| + |{}| − |{d, e, g, h}| =
3 + 0− 4 = −1 6≥ 0, and therefore S is not secure. 4

Encoding 1: Secure Sets (Loop Encoding πloop)

inS(V )← vertex(V ),not outS(V ). (1)

outS(V )← vertex(V ),not inS(V ). (2)

← #count{V : inS(V )} = 0. (3)

attackSet(V )← inS(U), edge(U, V ), outS(V ). (4)

border(U)← inS(U), outS(V ), edge(U, V ). (5)

inX(V ) | outX(V )← border(V ). (6)

outX(V )← inS(V ),not border(V ). (7)

defendSet(V )← inX(V ). (8)

defendSet(V )← inX(U), edge(U, V ), inS(V ). (9)

okupto(U, V )← vertex(U ;V ), inf(V ), outX(V ). (10)

okupto(U, V )← vertex(U ;V ), inf(V ), outS(V ). (11)

okupto(U, V )← vertex(U ;V ), inf(V ), inX(V ),not edge(U, V ). (12)

okupto(U,W )← vertex(U ;V ;W ), okupto(U, V ), succ(V,W ), outX(W ). (13)

okupto(U,W )← vertex(U ;V ;W ), okupto(U, V ), succ(V,W ), outS(W ). (14)

okupto(U,W )← vertex(U ;V ;W ), okupto(U, V ), succ(V,W ),

inX(W ),not edge(U,W ).

(15)

ok(U)← okupto(U, V ), sup(V ). (16)

inactAtt(U)← ok(U), outS(U), edge(U, V ), inS(V ). (17)

defended← #sum{1, V, pos : defendSet(V );

1, V, pos : inactAtt(V );

−1, V, neg : attackSet(V )} ≥ 0.

(18)

inX(V )← defended, inS(V ). (19)

outX(V )← defended, inS(V ). (20)

← not defended. (21)

3.2 Alternative Secure Set Characterization

In order to avoid the loop from above, we provide an alternative characterization
for secure sets, which to the best of our knowledge has not appeared in the
literature yet. The intuition is to guess instead of the set X ⊆ S a partition of
the attack set N [S]\S into active and inactive attackers, and then check whether
for such a partition a suitable set X exists.



a b c d

e f g h i

inS (S)

defendSet (N [X] ∩ S) inX (X)

attackSet

(N [S]\S)

inactAtt ((N [S]\N [X])\S)

Fig. 1: Example graph Gex with S = {a, b, c, f} and X = {c}.

Definition 2. Let G = (V,E) and S ⊆ V . We call any subset B ⊆ N [S] \ S a
partial boundary of S and identify by χ(B) = {X ⊆ S | (N [X] \ S) = B} the
subsets of S that cover B. B is called a valid partial boundary if χ(B) 6= ∅.

Lemma 3. Let G = (V,E) and S ⊆ V . Then S is secure in G iff for all valid
partial boundaries B and all X ∈ χ(B), |N [X] ∩ S| ≥ |B|.

Proof. “If” direction: Suppose S is not secure. Then there exists X ⊆ S, such
that |N [X] ∩ S| < |N [X] \ S|. Let B = N [X] \ S. We show that B is indeed a
valid partial boundary. Obviously, B ⊆ (N [S] \ S). Furthermore, X ∈ χ(B) and
by assumption |N [X] ∩ S| < |B|. “Only if” direction: Suppose there is a valid
partial boundary B and an X ∈ χ(B) such that |N [X]∩S| < |B|. By definition
X ⊆ S and B = N [X] \ S. Then S is not secure. ut

We can strengthen the result as follows.

Definition 3. Let G = (V,E) and S ⊆ V , and B ⊆ N [S] \ S. Moreover, let
b(S) be the border of S. Then χ′(B) = {X ⊆ b(S) | (N [X] \ S) = B}.

Lemma 4. Let G = (V,E) and S ⊆ V . Then S is secure in G iff for all valid
partial boundaries B and all X ∈ χ′(B), |N [X] ∩ S| ≥ |B| holds.

Proof. The “only if” direction follows directly from Lemma 3 and the fact that
χ′(B) ⊆ χ(B) for all partial boundaries B of S. For the other direction note that
|N [X] ∩ S| ≥ |B| holds for X ∈ χ(B) \ χ′(B) since for each Y ∈ χ(B) there is
an Y ′ ∈ χ′(B) with Y ′ ⊆ Y and that in this case |N [Y ] ∩ S| ≥ |N [Y ′] ∩ S|. ut

The main advantage of this definition is that we have for each relevant X ⊆ S,
i.e., for each X ⊆ b(S), the set (N [S] \ N [X]) \ S directly given via N [S] \ B
where B is some partial boundary of S with X ∈ χ(B).

In Encoding 2 we again first guess S (rules 1–3), and obtain the (directed)
edges from S to N [X]\S, denoted by borderEdge/2, as well as the attackSet/1
(rules 4–5). Next, the partial boundaries B ⊆ N [S]\S, or active attackers (de-
noted by actAtt/1), are guessed (rule 6). In order to obtain the sets X where
χ′(B) = {X ⊆ b(S) | (N [X] \ S) = B}, χ′ 6= ∅ holds, each vertex in B must



be adjacent to at least one vertex in X ⊆ b(s), denoted by inX/1 (rule 7). Fur-
thermore, if there is a vertex v ∈ N [X]\S with v /∈ B, then we can set X to be
defended, since X /∈ χ′(B) but X ∈ χ′(B′) with B ⊂ B′ (rule 8). Note that this
is required to be able to saturate. Rules 9–11 of Encoding 2 correspond to rules
8, 9 and 18 of Encoding 1, i.e., they define the defend set and check whether the
inequality given in Lemma 2 holds. Analogous to the loop encoding, we saturate
on the guessed inX/1, actAtt/1 and inactAtt/1 predicates. Due to Lemma 4,
we thus obtain the secure sets.

Encoding 2: Secure Sets (Alternative Encoding πalt)

inS(V )← vertex(V ),not outS(V ). (1)

outS(V )← vertex(V ),not inS(V ). (2)

← #count{V : inS(V )} ≤ 0. (3)

borderEdge(U, V )← inS(U), outS(V ), edge(U, V ). (4)

attackSet(V )← borderEdge(U, V ). (5)

actAtt(V ) | inactAtt(V )← attackSet(V ). (6)

inX(U) : borderEdge(U, V )← actAtt(V ), outS(V ). (7)

defended← inactAtt(U), inX(V ), edge(U, V ). (8)

defendSet(V )← inX(V ). (9)

defendSet(V )← inX(U), edge(U, V ), inS(V ). (10)

defended← #sum{1, V, pos : defendSet(V );

1, V, pos : inactAtt(V );

−1, V, neg : attackSet(V )} ≥ 0.

(11)

inX(V )← defended, inS(V ). (12)

actAtt(V )← defended, attackSet(V ). (13)

inactAtt(V )← defended, attackSet(V ). (14)

← not defended. (15)

3.3 Problem Variants and Optimizations

Optimization problem. In order to solve the Minimum Secure Set problem, it
is sufficient to add the minimize statement

#minimize{1, X, inS : inS(X), vertex(X)}.

to πloop and πalt. With this, only cardinality-minimal secure sets are obtained,
i.e., the answer sets with a minimal number of inS/1 occurrences are returned.

Counting alternative. The encodings presented so far rely on the #sum aggregate
for checking whether a set X is defended (see πloop rule 18, and πalt rule 11). The



#sum aggregate is required because the size of the attack set is subtracted from
the combined size of the defend and inactive attacker sets. By adding |V | to both
sides of the inequality in Lemma 2, and since |V | − |N [S]\S| = |V \(N [S]\S)|,
we get rid of the negative part in the inequality. Thus, it is possible to use
the #count aggregate by replacing rule 18 (πloop) and rule 11 (πalt) with the
following rules:

size(N)← N = #count{V : vertex(V )}.
defended← #count{V, pos : defendSet(V );

V, pos : inactAtt(V );

V, neg : vertex(V ),not attackSet(V )} ≥ N, size(N).

Note that default negation is allowed here, since attackSet/1 is independent of
the saturation. Furthermore, instead of adding |V |, |N [S]\S| could also simply
be shifted to the right side of the inequality. However, this drastically increases
the size of the grounding.

Search-space pruning. In order to improve the performance of our encodings,
it is desirable to define “strengthening” constraints. Here, we define constraints
that kill insecure sets S ⊆ V , independent of the guess for X ⊆ S. A vertex
v ∈ S cannot be defended if |N(v) ∩ S| < b|N(v)|/2c, or, in other words, less
than half of its neighbors are in S. In that case S cannot be secure. In ASP, this
is represented as follows:

deg(V,D)← vertex(V ), D = #count{U : edge(U, V )}.
← inS(V ), deg(V,D),#count{U : edge(U, V ), inS(U)} < D/2.

Note that degree deg/2 can be computed during grounding (i.e., it is independent
of any guess). Furthermore, ASP implements integer arithmetic, therefore D/2
always corresponds to b|N(v)|/2c.

4 Experimental Results

In this section we present the results of a preliminary performance analysis. In
detail, we compare our original Loop Encoding (which is not presented in this
paper), the Loop Encoding with Restriction to Border Vertices (Encoding 1)
and the Alternative Encoding (Encoding 2) for both enumerating all secure sets
and for enumerating all minimum secure sets. For all the encodings under inves-
tigation we use the search-space pruning approach described above to increase
performance. Furthermore, for performance reasons, we replace rules 1-3 of the
encodings with an appropriate choice rule. We omit a discussion of the count-
ing alternative here as our first experiments showed no significant difference in
performance between using #sum and #count in the encodings.



4.1 Benchmark Setup

We evaluate our encodings based on a set of graphs (generated using the Erdös-
Rényi random graph model) of different sizes n between 12 and 20 vertices. The
graphs were generated using a fixed edge probability p (i.e., the probability of
adding an edge between a pair of vertices). In our tests, we investigated the
influence of the edge probability for a range between 20 and 100 percent. This
allows us to analyze the impact of the size of the input graph and the impact
of the graph density on the overall performance of our encodings separately. For
each edge probability and for each graph size, 20 instances were generated and
tested. The benchmark results were obtained using a machine with two Intel
Xeon E5345 @ 2.33GHz processors and 48 GB RAM running openSUSE 11.4
x64. Each test run, using Clingo 4.3.0 [16], was limited to a single core and 256
MB RAM with a time limit of 15 minutes. Due to the fact that preliminary
tests showed no significant variance between the results of repeatedly executed
experiments, we only used one test run for each instance and encoding to obtain
the computation times which are represented in Figure 2.

4.2 Benchmark Discussion

Performance comparison of the different encodings. When interpreting Figure 2,
one can see that the original loop encoding and the loop encoding with restriction
to the border vertices are almost equally efficient while the alternative encoding
outperforms them in every case. We assume that the reason why the restriction to
border vertices does not show a significant performance improvement is the fact
that rules 4-6 of Encoding 1 introduce a kind of indirection that mitigates the
desired performance gain. In contrast to this, the strengths of the alternative
encoding (where, like in Encoding 1, only the border vertices are considered
within the subset check) are visible over the whole range of instances under
investigation and we assume that this also holds for larger graphs.

Influence of the graph size on computation time. Figures 2a and 2c show the
program’s runtime in the presence of different graph sizes. In particular note the
exponential explosion of the computation time in relation to the graph size for
both problem variants. Already for 22 vertices the time limit of 15 minutes was
exceeded by almost every instance. For the optimization problem, the variation
in runtime is much higher as the structure of the graph instance plays an im-
portant role (see also the discussion of the influence of the edge probability on
the program runtime right below). Figures 2a and 2c only cover the experiments
for instances of edge probability 0.5, but these observations also apply to other
edge probabilities which are not depicted here.

Influence of the edge probability on computation time. Figures 2b and 2d illus-
trate the strong dependency of the computation time on the edge probability
for a fixed graph size of 20 vertices. We assume that this can be explained by
the fact that “small” solutions (i.e., secure sets which contain a small number
of vertices) become less likely when the degree of connectedness increases. Inter-
estingly, the optimization encodings are much more sensitive to changes in the

Fig. 2: Performance results. The following abbreviations are used for the
different encoding variants: L - ”Loop Encoding”, B - ”Loop Encoding with

Restriction to the Border Vertices” and A - ”Alternative Encoding”.

edge probability than the enumeration variant. This is because the solver does
not have to check any larger candidates for secure sets in the encodings of the
optimization problem after a solution of smaller size has been found, while it
still has to check the whole search space when solving the enumeration variant.
Finally, note that for p = 1 all generated instances represent the same graph
(i.e. a clique), which explains the very small difference in runtime.

5 Conclusion

In this paper we have studied the problem of finding secure sets in graphs, which
is relevant, for example, in the context of studying relations between groups of
people. We have presented some alternative characterizations of secure sets that
we put to work in an ASP-based implementation.

From an ASP modeling point of view, finding secure sets in graphs is an
interesting problem as it requires disjunction as well as aggregates. It is therefore



an attractive candidate for testing the performance of ASP systems that allow
solving of problems harder than NP. In this work we have presented different
encodings that we believe to be useful for benchmarking ASP systems in the
future. This is of particular interest because so far problems harder than NP are
underrepresented among common collections of benchmark problems.

Recursive aggregates have posed a challenge in writing encodings for the
secure set problem. Our work witnesses that even some natural problems seem
to require relatively involved tricks in order to properly encode the required
arithmetic by means of aggregates in ASP. Moreover, current restrictions in the
DLV solver prevented us from gathering experimental data for this system.

Our experiments show a strong dependency of the execution time on the
graph size and the edge probability for all of our encodings, and the problem of
enumerating all secure sets in a graph is shown to be very challenging for today’s
ASP solvers even for relatively small instances. Our encoding that refrains from
looping over vertices (Encoding 2) outperforms our other encodings in all cases
and is therefore assumed to be a good starting point for further investigations.

We hope that in future versions of ASP systems restrictions with respect
to (recursive) aggregates are alleviated allowing further research on the secure
set problem as well as on similar problems requiring the full power of ASP.
In future work, we want to perform further analyses for such problems, which
should include other ASP systems (such as DLV) in order to obtain a more
comprehensive picture of how ASP can cope with them.

